

NAG Fortran Library Routine Document

F07TEF (STRTRS/DTRTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07TEF (STRTRS/DTRTRS) solves a real triangular system of linear equations with multiple right-hand sides, $AX = B$ or $A^T X = B$.

2 Specification

```

SUBROUTINE F07TEF(UPLO, TRANS, DIAG, N, NRHS, A, LDA, B, LDB, INFO)
ENTRY      strtrs (UPLO, TRANS, DIAG, N, NRHS, A, LDA, B, LDB, INFO)
INTEGER    N, NRHS, LDA, LDB, INFO
real     A(LDA,*), B(LDB,*)
CHARACTER*1 UPLO, TRANS, DIAG

```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine solves a real triangular system of linear equations $AX = B$ or $A^T X = B$.

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Higham N J (1989) The accuracy of solutions to triangular systems *SIAM J. Numer. Anal.* **26** 1252–1265

5 Parameters

1: UPLO – CHARACTER*1 *Input*

On entry: indicates whether A is upper or lower triangular as follows:

if UPLO = 'U', A is upper triangular;

if UPLO = 'L', A is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: TRANS – CHARACTER*1 *Input*

On entry: indicates the form of the equations as follows:

if TRANS = 'N', then the equations are of the form $AX = B$;

if TRANS = 'T' or 'C', then the equations are of the form $A^T X = B$.

Constraint: TRANS = 'N', 'T' or 'C'.

- 3: DIAG – CHARACTER*1 Input
On entry: indicates whether A is a non-unit or unit triangular matrix as follows:
 if DIAG = 'N', A is a non-unit triangular matrix;
 if DIAG = 'U', A is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.
Constraint: DIAG = 'N' or 'U'.
- 4: N – INTEGER Input
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 5: NRHS – INTEGER Input
On entry: r , the number of right-hand sides.
Constraint: NRHS ≥ 0 .
- 6: A(LDA,*) – *real* array Input
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the n by n triangular matrix A . If UPLO = 'U', A is upper triangular and the elements of the array below the diagonal are not referenced; if UPLO = 'L', A is lower triangular and the elements of the array above the diagonal are not referenced. If DIAG = 'U', the diagonal elements of A are not referenced, but are assumed to be 1.
- 7: LDA – INTEGER Input
On entry: the first dimension of the array A as declared in the (sub)program from which F07TEF (STRTRS/DTRTRS) is called.
Constraint: LDA $\geq \max(1, N)$.
- 8: B(LDB,*) – *real* array Input/Output
Note: the second dimension of the array B must be at least $\max(1, \text{NRHS})$.
On entry: the n by r right-hand side matrix B .
On exit: the n by r solution matrix X .
- 9: LDB – INTEGER Input
On entry: the first dimension of the array B as declared in the (sub)program from which F07TEF (STRTRS/DTRTRS) is called.
Constraint: LDB $\geq \max(1, N)$.
- 10: INFO – INTEGER Output
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO = i , a_{ii} is zero and the matrix A is singular.

7 Accuracy

The solutions of triangular systems of equations are usually computed to high accuracy. See Higham (1989).

For each right-hand side vector b , the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

$$|E| \leq c(n)\epsilon|A|,$$

$c(n)$ is a modest linear function of n , and ϵ is the *machine precision*.

If \hat{x} is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_{\infty}}{\|x\|_{\infty}} \leq c(n) \operatorname{cond}(A, x)\epsilon, \quad \text{provided } c(n) \operatorname{cond}(A, x)\epsilon < 1,$$

where $\operatorname{cond}(A, x) = \| |A^{-1}| |A| |x| \|_{\infty} / \|x\|_{\infty}$.

Note that $\operatorname{cond}(A, x) \leq \operatorname{cond}(A) = \| |A^{-1}| |A| \|_{\infty} \leq \kappa_{\infty}(A)$; $\operatorname{cond}(A, x)$ can be much smaller than $\operatorname{cond}(A)$ and it is also possible for $\operatorname{cond}(A^T)$ to be much larger (or smaller) than $\operatorname{cond}(A)$.

Forward and backward error bounds can be computed by calling F07THF (STRRFS/DTRRFS), and an estimate for $\kappa_{\infty}(A)$ can be obtained by calling F07TGF (STRCON/DTRCON) with NORM = 'I'.

8 Further Comments

The total number of floating-point operations is approximately n^2r .

The complex analogue of this routine is F07TSF (CTRTRS/ZTRTRS).

9 Example

To solve the system of equations $AX = B$, where

$$A = \begin{pmatrix} 4.30 & 0.00 & 0.00 & 0.00 \\ -3.96 & -4.87 & 0.00 & 0.00 \\ 0.40 & 0.31 & -8.02 & 0.00 \\ -0.27 & 0.07 & -5.95 & 0.12 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} -12.90 & -21.50 \\ 16.75 & 14.93 \\ -17.55 & 6.33 \\ -11.04 & 8.09 \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F07TEF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5, NOUT=6)
INTEGER          NMAX, LDA, NRHMAX, LDB
PARAMETER       (NMAX=8, LDA=NMAX, NRHMAX=NMAX, LDB=NMAX)
CHARACTER       TRANS, DIAG
PARAMETER       (TRANS='N', DIAG='N')
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N, NRHS
CHARACTER       UPLO
*      .. Local Arrays ..
real           A(LDA,NMAX), B(LDB,NRHMAX)
*      .. External Subroutines ..
EXTERNAL        strtrs, X04CAF
```

```

*      .. Executable Statements ..
WRITE (NOUT,*) 'F07TEF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N, NRHS
IF (N.LE.NMAX .AND. NRHS.LE.NRHMAX) THEN
*
*      Read A and B from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
      END IF
      READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*      Compute solution
*
      CALL strtrs(UPLO,TRANS,DIAG,N,NRHS,A,LDA,B,LDB,INFO)
*
*      Print solution
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
        IFAIL = 0
        CALL X04CAF('General',' ',N,NRHS,B,LDB,'Solution(s)',IFAIL)
      ELSE
        WRITE (NOUT,*) 'A is singular'
      END IF
      END IF
      STOP
*
      END

```

9.2 Program Data

F07TEF Example Program Data

```

4 2          :Values of N and NRHS
'L'         :Value of UPLO
4.30
-3.96 -4.87
0.40 0.31 -8.02
-0.27 0.07 -5.95 0.12 :End of matrix A
-12.90 -21.50
16.75 14.93
-17.55 6.33
-11.04 8.09      :End of matrix B

```

9.3 Program Results

F07TEF Example Program Results

Solution(s)

	1	2
1	-3.0000	-5.0000
2	-1.0000	1.0000
3	2.0000	-1.0000
4	1.0000	6.0000
